

Mathematical Exploration of Kirkwood Gaps

Jan Vrbik

We first solve the planar Kepler problem of an asteroid's motion, perturbed by the gravitational pull of Jupiter. Analyzing the resulting differential equations for its orbital elements, we demonstrate the mechanism for creating a gap at the 2:1 resonance (the asteroid making two orbits for Jupiter's one), and briefly mention the case of other resonances (3:2, 3:1, etc.). We also discuss reasons why the motion becomes chaotic at these resonances.

■ Planar Kepler Problem

Our aim is to solve the equation

$$\ddot{\mathbf{r}} + \mu \frac{\mathbf{r}}{r^3} = \epsilon \mathbf{f}, \quad (1)$$

where μ is the Sun's mass multiplied by the gravitational constant, \mathbf{r} is the asteroid's two-dimensional location (we represent vectors as complex quantities; r is the corresponding length), and $\epsilon \mathbf{f}$ is Jupiter's perturbing force, in the simplest form equal to

$$\epsilon \mu \left(\frac{R e^{i\lambda t} - \mathbf{r}}{|R e^{i\lambda t} - \mathbf{r}|^3} - \frac{e^{i\lambda t}}{R^2} \right). \quad (2)$$

Here, $\epsilon \approx 0.001$ is Jupiter's mass (relative to the Sun's mass) and $R e^{i\lambda t}$ is Jupiter's location (relative to the Sun's center). At this point, we take Jupiter's orbit to be a perfect circle of radius R in the plane of the asteroid's orbit and λ to be its constant angular speed (see [1]).

To solve (1), we introduce a new dependent variable U , and a new independent variable s , by

$$\mathbf{r} = U^2, \quad \frac{dt}{ds} = 2r \sqrt{\frac{a}{\mu}}, \quad (3)$$

where a is a positive (and at this point arbitrary) function of s . This definition implies that $r = U \bar{U}$, where \bar{U} denotes the complex conjugate of U .

The original equation now reads

$$\left(U'' - \frac{a'}{2a} U' \right) \bar{U} - \bar{U}' U' + 2a = 2\epsilon a \mathbf{f} U \bar{U}^3, \quad (4)$$

where the prime indicates differentiation with respect to s .

We verify this as follows.

$$\begin{aligned} \mathcal{D}[\mathbf{w}, \mathbf{t}] &:= \frac{\mathcal{D}[\mathbf{w}, \mathbf{s}]}{2 \mathbf{U}[\mathbf{s}] \bar{\mathbf{U}}[\mathbf{s}] \sqrt{a[\mathbf{s}] / \mu}}; \\ &\left(\mathcal{D}[\mathcal{D}[\mathbf{U}[\mathbf{s}]^2, \mathbf{t}], \mathbf{t}] + \frac{\mu}{\mathbf{U}[\mathbf{s}] \bar{\mathbf{U}}[\mathbf{s}]^3} - \epsilon \mu \mathbf{f} \right) 2 a[\mathbf{s}] \mathbf{U}[\mathbf{s}] \bar{\mathbf{U}}[\mathbf{s}]^3 / \mu // \\ &\text{Expand} \\ &2 a[\mathbf{s}] - 2 \mathbf{f} \epsilon a[\mathbf{s}] \mathbf{U}[\mathbf{s}] \bar{\mathbf{U}}[\mathbf{s}]^3 - \\ &\frac{\bar{\mathbf{U}}[\mathbf{s}] a'[\mathbf{s}] \mathbf{U}'[\mathbf{s}]}{2 a[\mathbf{s}]} - \mathbf{U}'[\mathbf{s}] \bar{\mathbf{U}}'[\mathbf{s}] + \bar{\mathbf{U}}[\mathbf{s}] \mathbf{U}''[\mathbf{s}] \end{aligned}$$

■ Unperturbed Solution

It is easy to show that the general solution to (4), when $\mathbf{f} = 0$, is

$$U = \sqrt{\frac{a}{1 + \beta^2}} \left(e^{i(s-s_0)} + \beta e^{-i(s-s_0)} \right) e^{i\phi/2}, \quad (5)$$

where a , β , s_0 , and ϕ are arbitrary constants (subject to $a > 0$ and $0 \leq \beta < 1$), called *orbital elements*. We next verify that this solution satisfies equation (4) (simplified, since now $a' = 0$ and $\mathbf{f} = 0$).

```

Module[{U},
  U = Sqrt[a/(1 + β^2)] (Exp[I (s - p)] + β Exp[-I (s - p)]) Exp[I φ / 2];
  Assuming[a > 0,
    ComplexExpand[Conjugate[U] D[U, {s, 2}] -
      D[U, s] Conjugate[D[U, s]] + 2 a] // Simplify]
]
0

```

Squaring U yields

$$\mathbf{r} = \frac{a}{1 + \beta^2} \left(e^{2i(s-s_0)} + 2\beta + \beta^2 e^{-2i(s-s_0)} \right) e^{i\phi}, \quad (6)$$

which is the usual ellipse (a is the length of its semimajor axis and $\frac{2\beta}{1+\beta^2}$ is its eccentricity), first stretched along the x axis and then rotated by the angle ϕ . The remaining orbital element s_0 is the value of s at aphelion.

■ Perturbed Solution

When \mathbf{f} is nonzero, we have to allow the orbital elements to be slowly varying functions of s and U itself to be extended to

$$U = \sqrt{\frac{a}{1 + \beta^2}} \left(q + \beta q^{-1} + (c_3 + i d_3) q^3 \right) e^{i\phi/2}, \quad (7)$$

where $c_3 + i d_3$ is a small complex number and $q \equiv e^{i(s-s_0)}$. In general, the big parentheses should contain terms with all odd powers of q (including negative ones), but this form is sufficient for our purpose.

Substituting trial solution (7) into the left-hand side of (4), discarding terms of the second and higher degrees in ϵ and β as too small, and collecting terms of the same degree in q , we get the following coefficients of q^{-2} , q^0 , and q^2 .

```

LHS = Module[{U, OverBar},
   $\epsilon^{i-}; i > 1 \wedge := 0;$ 
  Derivative[1][q] = Function[s, I (1 -  $\epsilon$  s0'[ $\epsilon$  s]) q[s]];
   $\overline{w} := w /. \{\text{Complex}[a_, b_] \rightarrow \text{Complex}[a, -b], q[s] \rightarrow q[s]^{-1}\};$ 
   $U = \sqrt{a[\epsilon s]} \left( 1 - \frac{1}{2} \beta[\epsilon s]^2 \right)$ 
  ( $q[s] + \beta[\epsilon s] / q[s] + \epsilon (c_3 + I d_3) q[s]^3$ ) Exp[I  $\phi[\epsilon s] / 2$ ];
  Table[Coefficient[
    ( $\overline{U} D[U, \{s, 2\}] + 2 a[\epsilon s] - D[U, s] \overline{D[U, s]} -$ 
 $\epsilon a'[\epsilon s] / 2 / a[\epsilon s] \overline{U} D[U, s]$ ) // Expand, q[s], i],
    {i, -2, 2, 2}] /.  $\epsilon \rightarrow 1 /. a[s] \rightarrow 1 /. \beta[s] \rightarrow \beta /.$ 
 $\beta^{i-}; i > 1 \rightarrow 0$ ];
  MatrixForm[LHS]

```

$$\begin{pmatrix} -4 c_3 + 4 i d_3 + \frac{1}{2} i \beta a'[s] - i \beta'[s] + \beta \phi'[s] \\ \frac{1}{2} i a'[s] + 4 s0'[s] - 4 i \beta \beta'[s] - 2 \phi'[s] \\ -12 c_3 - 12 i d_3 - \frac{1}{2} i \beta a'[s] - i \beta'[s] - \beta \phi'[s] \end{pmatrix}$$

These need to be matched against the coefficients of q^{-2} , q^0 , and q^2 obtained when the right-hand side of the equation is similarly expanded.

We now proceed to do just that.

□ Resonance Variable

To simplify subsequent computation, we use units that make both μ and a (when in the exact 2:1 resonance with Jupiter; see [2]) equal to 1. Referring to the perturbing force (2), this makes $R = 2^{2/3}$ and $\lambda = 1/2$; the new unit of time is roughly one year (0.944 years, to be exact). According to the second line of (3), we get, to sufficient accuracy (i.e. using the unperturbed solution and discarding higher powers of β),

$$\begin{aligned}
 t &= 2 \int \sqrt{a} U \overline{U} ds = 2 \int a^{3/2} (1 + 2 \beta \cos(2(s - s_0))) ds = \\
 &2s + 2 \int (a^{3/2} - 1) ds + 2 \beta \sin(2(s - s_0)) \equiv \\
 &2(s - s_0) + 2\phi - \Psi + 2\beta \sin(2(s - s_0)).
 \end{aligned} \tag{8}$$

The usual additive constant can always be eliminated by the corresponding choice of the s -scale origin.

The last equality defines the so-called “resonance” variable

$$\Psi = 2\phi - 2s_0 + 2 \int (a^{3/2} - 1) ds. \quad (9)$$

This implies that

$$\Psi' = 2\phi' - 2s_0' + 2(a^{3/2} - 1) \simeq 2\phi' - 2s_0' + 3(a - 1) \quad (10)$$

since $a \simeq 1$.

Replacing t by the last line of (8), Jupiter’s position is thus given by

$$R e^{i t/2} = R q e^{i \phi - i \Psi/2 + i \beta \sin(2(s-s_0))}, \quad (11)$$

where $R = 2^{2/3}$. We are now ready to evaluate the coefficients of q^{-2} , q^0 , and q^2 by expanding the right-hand side of (4).

```

RHS = Module[{U, J, f, OverBar},
  w_ := w /. {Complex[a_, b_] -> Complex[a, -b],
    q[s] -> q[s]^-1};
  U = (q[s] + beta / q[s]) Exp[I phi / 2];
  J = R q[s] Exp[I (phi - Psi / 2 + beta (q[s]^2 - q[s]^-2) / 2 / I)];
  f =
    Series[
      (
        (J - U^2) / ((J - U^2)^(3/2)) - J / R^3
      ) U^3 U {q[s]^2, 1, q[s]^-2},
      {beta, 0, 1}
    ] /. R -> 2^(2/3) /. q[s] -> Exp[I xi - I Psi / 2] //
    Normal // Simplify // PowerExpand;
  G[w_] := NIntegrate[w, {xi, 0, 2 pi}];
  Collect[f / pi // ExpToTrig // TrigExpand,
    {beta, Sin[Psi], Cos[Psi]}, G] // Simplify // Chop
]

{
  (Cos[Psi] - I Sin[Psi])
  (1.03966 + 8.79527 beta Cos[Psi] - (0. + 4.88823 I) beta Sin[Psi]),
  0.438956 + 8.30394 beta Cos[Psi] - (0. + 5.99971 I) beta Sin[Psi],
  (Cos[Psi] + I Sin[Psi])
  (0.119127 + 2.52099 beta Cos[Psi] - (0. + 1.38605 I) beta Sin[Psi])
}

```

□ Resulting Equations

Matching the coefficients of q^{-2} , q^0 , and q^2 between the left- and right-hand sides of (4) yields three complex (six real) linear equations for a' , β' , ϕ' , s_0' , c_3 , and d_3 . These can be easily solved, resulting in the following three expressions for a' , β' , and Ψ' , respectively.

```
Module[{sol},
  sol =
    NSolve[Flatten[{Thread[ComplexExpand[Re[LHS] == Re[RHS]]],
      Thread[ComplexExpand[Im[LHS] == Im[RHS]]]}],
      {a'[s], beta'[s], phi'[s], s0'[s], c3, d3}][[1]] //
    Simplify // Chop;
  {
    Series[a'[s] /. sol, {beta, 0, 1}] // Simplify,
    Series[beta'[s] /. sol, {beta, 0, 0}],
    Series[2 phi'[s] - 2 s0'[s] /. sol, {beta, 0, -1}] // Chop
  } // Column
]

-5.99971 Sin[Psi] beta + O[beta]^2
0.749964 Sin[Psi] + O[beta]^1
0.749964 Cos[Psi]
beta + O[beta]^0
```

Only the leading term of the corresponding β -expansion has been kept in each case. One can show that the additional terms would only minutely affect the resulting solution.

■ Dynamics of 2:1 Resonance

The forced changes of the asteroid's orbital elements when in or near the 2:1 resonance are thus described to sufficient accuracy by the following three differential equations

$$\begin{aligned} a' &= -6\epsilon \beta \sin \Psi, \\ \beta' &= 0.75\epsilon \sin \Psi, \\ \Psi' &= \frac{0.75\epsilon}{\beta} \cos \Psi - 3(a - 1). \end{aligned} \tag{12}$$

Clearly, $a' + 8\beta\beta' = 0$, which implies that $a + 4\beta^2 \equiv 1 + K$ is a constant of motion. Multiplying the last equation of (12) by β , replacing a by $1 + K - 4\beta^2$, and collecting all terms on the right-hand side yields

$$0 = 0.75\epsilon \cos \Psi - 3K\beta + 12\beta^3 - \beta\Psi'. \tag{13}$$

Multiplying each term by β' (or, equivalently, by $0.75 \epsilon \sin \Psi$) results in


$$0 = 0.75 \epsilon \beta' \cos \Psi - 3 K \beta \beta' + 12 \beta^3 \beta' - 0.75 \epsilon \sin \Psi \beta \Psi', \quad (14)$$

which makes it obvious that

$$0.75 \epsilon \beta \cos \Psi - \frac{3 K}{2} \beta^2 + 3 \beta^4 \quad (15)$$

is another constant of motion. Displaying the corresponding contours when $K = 0.016$ illustrates that there are four distinct types of solution to (12).

```
ContourPlot[.00075  $\beta$  Cos[ $\Psi$ ] - 3 / 2  $K$   $\beta^2$  + 3  $\beta^4$  / .  $K$  -> .016,  
{ $\Psi$ , -.5, 4}, { $\beta$ , .0, .08}, ContourShading -> False,  
PlotPoints -> 150, Contours -> 70, FrameLabel -> { $\Psi$ ,  $\beta$ }]
```




Specifically, there are two centers (at $\Psi = 0$ and $\Psi = \pi$, of low and high eccentricity, respectively) and their basins, with the resonance variable librating. Between these two basins, and then again at high eccentricities, there are two regions (separated by a hyperbolic fixed point) where Ψ circulates.

In all four cases β , and correspondingly a , oscillate in a regular manner; there is no tendency to “clear” the resonance region. To achieve just that, an extra perturbation is required.

Before bringing it in, another important fact needs to be mentioned: the previous four-region solution occurs only when $K > 0.0119$. As soon as K reaches this “critical value,” the $\Psi = 0$ center and the hyperbolic fixed point merge into one, and then (for $K < 0.0119$) both disappear, leaving only the $\Psi = \pi$ basin and the high-eccentricity solutions, as seen in the following plot.

```
ContourPlot[.00075  $\beta$  Cos [ $\Psi$ ] - 3 / 2  $\kappa$   $\beta^2$  + 3  $\beta^4$  / .  $\kappa$  -> .0119,  
{ $\Psi$ , -.5, 4}, { $\beta$ , .0, .08}, ContourShading -> False,  
PlotPoints -> 150, Contours -> 70, FrameLabel -> { $\Psi$ ,  $\beta$ }]
```



This helps to understand the actual mechanism of clearing the gap, which is discussed next.

□ Kepler Shear

All orbiting bodies are constantly bombarded by celestial debris (meteoroids and such). When the body (such as an asteroid) is relatively small, this may affect its orbit, however slightly, due to the following effect: at aphelion, the asteroid is moving rather slowly compared to nearby objects, and is more likely to be hit from behind; near perihelion, it is the exact opposite. It can be shown (see [3]) that this will add a term proportional to $i \beta^2 (3 q^{-2} - q^2)$ to the right-hand side of (4), consequently modifying the expression for β' , namely


$$\beta' = 0.75 \epsilon \sin \Psi - C \beta^2, \quad (16)$$

where C is a small constant. To verify this, in the program of the Resulting Equations section, add $C \text{ I } \beta^2 \{3, 0, -1\}$ (our code for $C i \beta^2 (3 q^{-2} - q^2)$) to RHS and recompute β' , allowing for the extra powers of β .

The new term in (16) will modify the corresponding solution to (12) in the following ways:

- Each center becomes attracting (a solution within its basin will slowly spiral toward the center).
- K will (more slowly yet) decrease, until the center at $\Psi = 0$ disappears. At that point, a low-eccentricity solution is suddenly converted into a high-eccentricity orbit, with the corresponding sudden decrease in the value of a , as demonstrated next.

```
Module[
  { $\epsilon = .001$ , sol},
  BlockRandom[
    SeedRandom[245];
    sol = NDSolve[{a'[s] == -6  $\epsilon \beta[s] \sin[\Psi[s]]$ ,
       $\beta'[s] == .75 \epsilon \sin[\Psi[s]] - .0005 \beta[s]^2$ ,
       $\Psi'[s] == .75 \epsilon / \beta[s] \cos[\Psi[s]] - 3 (a[s] - 1)$ ,
      a[0] == 1.01 + RandomReal[0.01],  $\beta[0] == \text{RandomReal}[0.04]$ ,
       $\Psi[0] == \text{RandomReal}[2 \pi]$ }, {a,  $\beta$ ,  $\Psi$ }, {s, 0, 200 000},
      MaxSteps -> 300 000]
  ];
  Plot[a[s] /. sol[[1]], {s, 0, 200 000}, PlotPoints -> 200,
    AxesLabel -> {s, a}]
]
```



Since the initial values of a , β , and Ψ are generated randomly, they will occasionally (after removing `SeedRandom[245]`) result in starting inside (or below) the gap. Nevertheless, by running the program several times, one can verify that no initial condition now allows a to stay inside the gap (meaning, roughly, $0.99 < a < 1.01$).

□ Other Perturbations

There are clearly many other perturbations acting on an asteroid beyond the two we have considered so far (Jupiter's gravity and Kepler shear). Rather than considering them individually, we will combine them into a single new term added to the right-hand side of the first equation in (12), since such a term is most effective in visibly affecting the nature of the previous solution. To simplify matters, we make the new term a small constant, even though in reality it may have a slow time dependence. The new equation then reads


$$a' = -6\epsilon\beta\sin\Psi + \kappa. \quad (17)$$

When $\kappa > 0$, the old solution is not much affected, only the sudden crossing of the gap becomes a touch faster. Things change dramatically when $\kappa < 0$; in this case, there is (when $C \cdot \kappa \ll \epsilon^2$ and $\kappa \ll C$) a fixed point below the gap to which most solutions are drawn (except for initial values of $a > 1.01$; these are outside the basin of its attraction, and the corresponding solutions just slowly drift away from the gap). We can demonstrate this by running the following program several times to explore all possibilities (again, after removing `SeedRandom[3]`).

```

Module[
  { $\epsilon$  = .001, sol},
  BlockRandom[
    SeedRandom[3];
    sol = NDSolve[{a'[s] == -6  $\epsilon$   $\beta$ [s] Sin[ $\Psi$ [s]] + 0.000001,
       $\beta$ '[s] == .75  $\epsilon$  Sin[ $\Psi$ [s]] - .0005  $\beta$ [s]2,
       $\Psi$ '[s] == .75  $\epsilon$  /  $\beta$ [s] Cos[ $\Psi$ [s]] - 3 (a[s] - 1),
      a[0] == 0.97 + RandomReal[0.05],  $\beta$ [0] == RandomReal[0.05],
       $\Psi$ [0] == RandomReal[2  $\pi$ ]}, {a,  $\beta$ ,  $\Psi$ }, {s, 0, 300 000},
      MaxSteps → 400 000];
    Plot[a[s] /. sol[[1]], {s, 0, 300 000}, PlotPoints → 200,
      AxesLabel → {s, a}]
  ]
]

```



As already mentioned, the value of κ may, in reality, slowly change in time. But, as we have seen, regardless of its value and sign, a gap is always cleared.

■ Chaos

In a similar manner, one can show that, when Jupiter's eccentricity ($\gamma \simeq 0.05$) is accounted for, ϕ appears on the right-hand side of (12), implying that the three equations must be extended to the following four:

$$\begin{aligned} a' &= -6\epsilon \beta \sin \Psi - 1.08\epsilon \gamma \sin(\Psi - \phi), \\ \beta' &= 0.75\epsilon \sin \Psi - \epsilon \gamma (3.13 \sin(2\Psi - \phi) - 0.36 \sin \phi), \\ \Psi' &= \epsilon \frac{0.75 \cos \Psi + \gamma(3.13 \cos(2\Psi - \phi) - 0.36 \cos \phi)}{\beta} - 3(a - 1), \\ \phi' &= \epsilon \frac{0.75 \cos \Psi + \gamma(3.13 \cos(2\Psi - \phi) - 0.36 \cos \phi)}{\beta}. \end{aligned} \quad (18)$$

At this point, we have not yet included Kepler's shear, nor any other additional perturbation.

Let us see how this changes the original, very regular solution near the 2:1 resonance.

```
Module[
  { $\epsilon$  = .001,  $\gamma$  = .05, sol},
  BlockRandom[
    SeedRandom[287];
    sol =
      NDSolve[
        {
          a'[s] == -6  $\epsilon$   $\beta$ [s] Sin[ $\Psi$ [s]] - 1.08  $\epsilon$   $\gamma$  Sin[ $\Psi$ [s] -  $\phi$ [s]],
           $\beta$ '[s] == .75  $\epsilon$  Sin[ $\Psi$ [s]] + 0.36  $\epsilon$   $\gamma$  Sin[ $\phi$ [s]] +
            3.13  $\epsilon$   $\gamma$  Sin[2  $\Psi$ [s] -  $\phi$ [s]],
           $\Psi$ '[s] ==
            
$$\frac{1}{\beta[s]} (.75 \epsilon \cos[\Psi[s]] + 0.36 \epsilon \gamma \cos[\phi[s]] +$$


$$3.13 \epsilon \gamma \cos[2 \Psi[s] - \phi[s]]) - 3(a[s] - 1),$$

           $\phi$ '[s] ==
            
$$\frac{1}{\beta[s]} (.75 \epsilon \cos[\Psi[s]] + 0.36 \epsilon \gamma \cos[\phi[s]] +$$



$$3.13 \epsilon \gamma \cos[2 \Psi[s] - \phi[s]]),$$

          a[0] == 0.99 + RandomReal[0.02],  $\beta$ [0] == RandomReal[0.1],
           $\Psi$ [0] == RandomReal[2  $\pi$ ],  $\phi$ [0] == RandomReal[2  $\pi$ ]
        },
        {a,  $\beta$ ,  $\Psi$ ,  $\phi$ }, {s, 0, 100 000}, MaxSteps -> 200 000
      ]
  ];
```

```

Column@{
  Plot[ $\beta[s]$  /. sol[[1]], {s, 0, 100 000}, ImageSize -> 360,
    PlotPoints -> 200, AxesLabel -> {s,  $\beta$ },
  Plot[a[s] /. sol[[1]], {s, 0, 100 000}, ImageSize -> 360,
    PlotPoints -> 200, AxesLabel -> {s, a}]
}
]

```



One can see that the solution, under these circumstances, occasionally becomes quite chaotic (very sensitive to initial conditions), in which case β can reach rather large values. Beyond that, nothing very interesting happens to a ; its value always remains in the resonance region. Clearly, chaos is not the main reason for clearing the gaps, as is often incorrectly stated.

The same conclusion can be reached when including *inclination* between the two orbital planes. In that case, a set of six differential equations is needed (one for the inclination and the other for the remaining Euler angle), and the chance of getting a chaotic solution slightly increases. But no gap clearing is ever observed (without the crucial extra perturbations of our previous solutions).

■ Other Resonances

For resonances of the type $n : (n - 1)$ (where n is a small integer), the resulting equations and the corresponding conclusions remain almost identical to those of the 2:1 case (only numerical coefficients differ). Thus, for example, for a motion near the 3:2 resonance, we get

$$\begin{aligned} a' &= -24.7 \epsilon \beta \sin \Psi, \\ \beta' &= 1.55 \epsilon \sin \Psi, \\ \Psi' &= \frac{1.55 \epsilon}{\beta} \cos \Psi - 6(a - 1). \end{aligned} \tag{19}$$

For $n : (n - 2)$ resonances, the situation is slightly more complicated, as higher powers of β become the leading terms on the right-hand side of each equation. We quote results for the important case of 3:1 resonance (also investigated in [4]).

$$\begin{aligned} a' &= -4.61 \epsilon \beta^2 \sin \Psi, \\ \beta' &= 1.15 \epsilon \beta \sin \Psi, \\ \Psi' &= 0.47 \epsilon + 2.30 \epsilon \cos \Psi - 3(a - 1). \end{aligned} \tag{20}$$

This trend (of increasing powers of β on the right-hand side of each equation, with the exception of Ψ') continues, as we move on to $n : (n - 3)$ and $n : (n - 4)$ resonances. Thus, for example, we get

$$\begin{aligned} a' &= -39.4 \epsilon \beta^3 \sin \Psi, \\ \beta' &= 7.38 \epsilon \beta^2 \sin \Psi, \\ \Psi' &= 1.16 \epsilon - 6(a - 1), \end{aligned} \tag{21}$$


for the 5:2 resonance, and

$$\begin{aligned} a' &= -245 \epsilon \beta^4 \sin \Psi, \\ \beta' &= 40.8 \epsilon \beta^3 \sin \Psi, \\ \Psi' &= 1.88 \epsilon - 9(a - 1), \end{aligned} \tag{22}$$

for 7:3.

This time (i.e. beyond the $n:(n-1)$ case), the exact mechanism for creating a gap is slightly different from the 2:1 case: when $\kappa > 0$, there is no longer any fixed point below the gap; instead, a makes a quick transition through the gap, and stabilizes its value above it, in a manner similar to the $\kappa < 0$ case (except for the reversal of direction), which we now demonstrate using the 7:3 resonance.

```
Module[
  { $\epsilon = .001$ , sol},
  BlockRandom[SeedRandom[7];
  sol = NDSolve[{a'[s] == -245  $\epsilon \beta[s]^4 \text{Sin}[\Psi[s]] + 0.00000001$ ,
     $\beta'[s] == 40.8 \epsilon \beta[s]^3 \text{Sin}[\Psi[s]] - .000001 \beta[s]^2$ ,
     $\Psi'[s] == 1.88 \epsilon - 9(a[s] - 1)$ , a[0] == 0.999,  $\beta[0] == 0.05$ ,
     $\Psi[0] == \text{RandomReal}[2 \pi]$ }, {a,  $\beta$ ,  $\Psi$ }, {s, 0, 100 000},
    MaxSteps -> 200 000];
  Plot[a[s] /. sol[[1]], {s, 0, 100 000}, PlotPoints -> 200,
    AxesLabel -> {s, a}]
]
```



In the general case of $n:(n-k)$ resonance, the gap becomes narrower with increasing k .

■ References

- [1] J. Vrbik, "Resonance Formation of Kirkwood Gaps and Asteroid Clusters," *Journal of Physics A: Mathematical and General*, **29**(12), 1996 pp. 3311–3316.
iopscience.iop.org/0305-4470/29/12/033.
- [2] M. Moons, A. Morbidelli, and F. Migliorini, "Dynamical Structure of the 2/1 Commensurability with Jupiter and the Origin of the Resonant Asteroids," *Icarus*, **135**(2), 1998 pp. 458–468.
doi:10.1006/icar.1998.5963.
- [3] J. Vrbik, "Perturbative Solution of the Motion of an Asteroid in Resonance with Jupiter," *Monthly Notices of the Royal Astronomical Society*, **316**(3), 2000 pp. 459–463.
onlinelibrary.wiley.com/doi/10.1046/j.1365-8711.2000.03450.x/abstract.
- [4] J. Wisdom, "A Perturbative Treatment of Motion Near the 3/1 Commensurability," *Icarus*, **63**(2), 1985 pp. 272–289. doi:10.1016/0019-1035(85)90011-9.

J. Vrbik, "Mathematical Exploration of Kirkwood Gaps," *The Mathematica Journal*, 2012.
dx.doi.org/doi:10.3888/tmj.14–1.

About the Author

Jan Vrbik

Department of Mathematics, Brock University
500 Glenridge Ave., St. Catharines
Ontario, Canada, L2S 3A1
jvrzik@brocku.ca